Cálculo Diferencial e Integral: um KIT de sobrevivência

Prof. Doherty Andrade

Transformada de Laplace

Teorema (Existência da Transformada de Laplace) Se $f(t)$ é de ordem exponencial, então sua transformada de Laplace $Lf(t) = F(s)$ é dada por

$$F(s) = \int_0^\infty f(t)\, e^{(-s\,t)}\, dt$$

.

A integral definida $F(s)$ existe nos pontos $\tau < s$.

Teorema (Linearidade da Transformada de Laplace) Sejam $f(t)$ e $g(t)$ tendo transformada de Laplace dadas por $F(s)$ e $G(s)$, respectivamente. Se a e b são constantes, então $L(a\, f(t) + b\, g(t)) = a\, F(s) + b\, G(s)$.

Teorema (Unicidade da Transformada de Laplace) Sejam $f(t)$ e $g(t)$ tendo Transformada de Laplace dadas por $F(s)$ e $G(s)$, respectivamente. Se $F(s) = G(s)$ então $f(t) = g(t)$ **.**

Para trabalhar com Transformada de Laplace no Maple, você precisa carregar os procedimentos "Laplace transform" .

Faça isto com

```
> with(inttrans):
```

Warning, new definition for hilbert

Exemplo 1 Determine a transformada de Laplace da função degrau unitário.

$f(t) = 1$ se $0 \le t < c$,

$f(t) = 0$ se $c < t$.

```
> c:='c': f:='f': F:='F': g:='g': s:='s': t:='t': T:='T':
f0 := t -> 1:
g := t -> subs(T=t, int(f0(T)*exp(-s*T),T)):
F := t -> subs(T=t, int(f0(T)*exp(-s*T),T=0..c)):
`For 0 <= t <= c, f(t) ` = f0(t);
Int(f(t)*exp(-s*t),t) = g(t);
`F(s) = `, Int(f(t)*exp(-s*t),t=0..c) = F(s);
`F(s) ` = simplify(F(s));
```

$$For\ 0 <= t <= c,\ f(t) = 1$$

$$\int f(t)\, e^{(-s\,t)}\, dt = -\frac{e^{(-s\,t)}}{s}$$

$$F(s) = ,\ \int_0^c f(t)\, e^{(-s\,t)}\, dt = -\frac{e^{(-s\,c)} - 1}{s}$$

$$F(s) = -\frac{e^{(-s\,c)} - 1}{s}$$

Veja o gráfico de f.

```
> f:=x -> piecewise(x>0,1);#tomei c=0 aqui
```

$$f := x \to \mathrm{piecewise}(0 < x, 1)$$

```
> with(plots):
```

```
> plot(f(x),x=-2..4,y=0..2);
```



Exemplo 2 Determine a transformada de Laplace de $f(t) = e^{(a\,t)}$.

```
> a:='a': f:='f': F:='F': g:='g': s:='s': t:='t': T:='T':
f0 := t -> exp(a*t):
`f(t) ` = f0(t);
g := proc(t,S)
simplify(subs(T=t,int(f0(T)*exp(-S*T),T)))
end:
Int(f(t)*exp(-s*t),t) = g(t,s);
`F(s) ` = subs(T=t, int(f0(T)*exp(-s*T),T=0..infinity));
`F(s) ` = simplify(g(infinity,s) - g(0,s));
F := s -> - subs(S=s, g(0,S)):
`F(s) ` = F(s);
```

$$f(t) = \mathbf{e}^{(a\,t)}$$

$$\int f(t)\,\mathbf{e}^{(-s\,t)}\,dt = \frac{\mathbf{e}^{(t\,(a-s))}}{a-s}$$

Definite integration: Can't determine if the integral is convergent.

Need to know the sign of --> -a+s

Will now try indefinite integration and then take limits.

$$F(s) = \lim_{t \to \infty} \frac{e^{(a\,t - s\,t)} - 1}{a - s}$$

$$F(s) = \frac{e^{(\infty\,(a - s))} - 1}{a - s}$$

$$F(s) = -\frac{1}{a - s}$$

```
> `f(t) ` = exp(a*t);
`F(s) ` = laplace(exp(a*t), t, s);
```

$$f(t) = e^{(a\,t)}$$

$$F(s) = \frac{1}{-a + s}$$

Exemplo 3 Determine a transformada de Laplace de $f(t) = \sinh(a\,t)$.

Como $f(t) = \sinh(at) = \dfrac{e^{(a\,t)} - e^{(-a\,t)}}{2}$, usamos que

$$L_1(e^{(a\,t)}) = \frac{1}{s - a}$$

e

$$L_2(e^{(-a\,t)}) = \frac{1}{s + a}$$

.

```
> L:='L':
`f(t) ` = sinh(a*t);
`f(t) ` = (exp(a*t)-exp(-a*t))/2;
L1 :=laplace( exp(a*t), t, s):
L2 :=laplace(exp(-a*t), t, s):
L(exp(a*t)) = L1;
L(exp(-a*t)) = L2; ` `;
`F(s) ` = (L(exp(a*t)) - L(exp(-a*t)))/2;
`F(s) ` = (L1 - L2)/2;
`F(s) ` = simplify((L1 - L2)/2);
```

$$f(t) = \sinh(a\,t)$$

$$f(t) = \frac{1}{2}\mathbf{e}^{(a\,t)} - \frac{1}{2}\mathbf{e}^{(-a\,t)}$$

$$\mathrm{L}(\mathbf{e}^{(a\,t)}) = \frac{1}{-a+s}$$

$$\mathrm{L}(\mathbf{e}^{(-a\,t)}) = \frac{1}{s+a}$$

$$F(s) = \frac{1}{2}\mathrm{L}(\mathbf{e}^{(a\,t)}) - \frac{1}{2}\mathrm{L}(\mathbf{e}^{(-a\,t)})$$

$$F(s) = \frac{1}{2}\frac{1}{-a+s} - \frac{1}{2}\frac{1}{s+a}$$

$$F(s) = -\frac{a}{(a-s)\,(s+a)}$$

Podemos verificar este resultado usando as rotinas do Maple.

```
> `f(t) ` = sinh(a*t);
`F(s) ` = laplace(sinh(a*t), t, s);
```

$$f(t) = \sinh(a\,t)$$

$$F(s) = \frac{a}{s^2 - a^2}$$

Exemplo 4 Determine a trasnformada de Laplace de $\mathrm{f}(t) = t$.

```
> a:='a': f:='f': F:='F': g:='g': s:='s': t:='t': T:='T':
f0 := t -> t:
`f(t) ` = f0(t);
g := proc(t,S)
simplify(subs(T=t,int(f0(T)*exp(-S*T),T)))
end:
Int(f(t)*exp(-s*t),t) = g(t,s);
`F(s) ` = subs(T=t, int(f0(T)*exp(-s*T),T=0..infinity));
`F(s) ` = simplify(g(infinity,s) - g(0,s));
```

F := s -> - subs(S=s, g(0,S)):
`F(s) ` = F(s);

$$f(t) = t$$

$$\int f(t)\, e^{(-s\,t)}\, dt = -\frac{e^{(-s\,t)}\,(s\,t+1)}{s^2}$$

Definite integration: Can't determine if the integral is convergent.

Need to know the sign of --> s

Will now try indefinite integration and then take limits.

$$F(s) = \lim_{t \to \infty} -\frac{s\,t\,e^{(-s\,t)} + e^{(-s\,t)} - 1}{s^2}$$

$$F(s) = -\frac{s\,\infty\,e^{(-s\,\infty)} + e^{(-s\,\infty)} - 1}{s^2}$$

$$F(s) = \frac{1}{s^2}$$

Podemos verificar este resultado usando as rotinas do pacote Transformada de Laplace.

> **`f(t) ` = t;**
`F(s) ` = laplace(t, t, s);

$$f(t) = t$$

$$F(s) = \frac{1}{s^2}$$

Exemplo 5 Determine a transformada de Laplace de $f(t) = \cos(b\,t)$.

> **a:='a': f:='f': F:='F': g:='g': s:='s': t:='t': T:='T':**
f0 := t -> cos(b*t):
`f(t) ` = f0(t);
g := proc(t,S)

```
simplify(subs(T=t,int(f0(T)*exp(-S*T),T)))
end:
Int(f(t)*exp(-s*t),t) = g(t,s);
`F(s) ` = subs(T=t, int(f0(T)*exp(-s*T),T=0..infinity));
`F(s) ` = simplify(g(infinity,s) - g(0,s));
F := s -> - subs(S=s, g(0,S)):
`F(s) ` = F(s);
```

$$f(t) = \cos(b\,t)$$

$$\int f(t)\, e^{(-s\,t)}\, dt = \frac{e^{(-s\,t)}\,(-s\cos(b\,t) + b\sin(b\,t))}{s^2 + b^2}$$

$$F(s) = \lim_{t \to \infty} \frac{-s\, e^{(-s\,t)} \cos(b\,t) + b\, e^{(-s\,t)} \sin(b\,t) + s}{s^2 + b^2}$$

$$F(s) = \frac{-s\, e^{(-s\,\infty)} \cos(b\,\infty) + b\, e^{(-s\,\infty)} \sin(b\,\infty) + s}{s^2 + b^2}$$

$$F(s) = \frac{s}{s^2 + b^2}$$

Verifique este resultado usando as rotinas do Maple.

```
> `f(t) ` = cos(b*t);
`F(s) ` = laplace(cos(b*t), t, s);
```

$$f(t) = \cos(b\,t)$$

$$F(s) = \frac{s}{s^2 + b^2}$$

Exemplo 6 Determine a transformada inversa de $F(s) = \dfrac{3\,s + 6}{s^2 + 9}$.

```
> f:='f': F:='F': s:='s': t:='t':
F0 := s -> (3*s + 6)/(s^2 + 9):
`F(s) ` = F0(s);
`F(s) ` = expand(F0(s));
```

$$F(s) = \frac{3s+6}{s^2+9}$$

$$F(s) = 3\frac{s}{s^2+9} + 6\frac{1}{s^2+9}$$

A transformada $\mathrm{F}(s)$ é uma combinação linear.

> **F1 := s/(s^2 + 9):**
F2 := 3/(s^2 + 9):
F[1](s) = F1;
F[2](s) = F2;
`F(s) = `, 3*F[1](s) + 2*F[2](s) = 3*F1 + 2*F2;

$$F_1(s) = \frac{s}{s^2+9}$$

$$F_2(s) = 3\frac{1}{s^2+9}$$

$$F(s) = , 3F_1(s) + 2F_2(s) = 3\frac{s}{s^2+9} + 6\frac{1}{s^2+9}$$

A inversa de $F_1(s)$ é $\cos(3t)$ e a inversa de $F_2(s)$ é $\sin(3t)$.

> **f1 := invlaplace(F1, s, t):**
f2 := invlaplace(F2, s, t):
f[1](t) , ` = L^-1 (F1(s)) ` = f1;
f[2](t) , ` = L^-1 (F2(s)) ` = f2;

$$f_1(t), \ = L\text{^-}1\ (F1(s)) \ = \cos(\sqrt{9}\,t)$$

$$f_2(t), \ = L\text{^-}1\ (F2(s)) \ = \frac{1}{3}\sqrt{9}\,\sin(\sqrt{9}\,t)$$

Portanto $\mathrm{f}(t)$ = $3f_1(t) + 2f_2(t)$ = $3\cos(3t) + 2\sin(3t)$.

> **`f(t) = `, 3*f[1](t) + 2*f[2](t) = 3*f1 + 2*f2;**

$$f(t) = , 3 f_1(t) + 2 f_2(t) = 3\cos(\sqrt{9}\,t) + \frac{2}{3}\sqrt{9}\,\sin(\sqrt{9}\,t)$$

Podemos verificar isto usando os procedimentos do Maple.

```
> `F(s) ` = (3*s + 6)/(s^2 + 9);
`f(t) ` = invlaplace((3*s + 6)/(s^2 + 9), s, t);
```

$$F(s) = \frac{3s+6}{s^2+9}$$

$$f(t) = 3\cos(3t) + 2\sin(3t)$$

Transformada de Laplace de derivadas e integrais

Teorema (Derivada de f(t)) Seja $f(t)$ e $f'(t)$ contínuas em $0 \le t$, e de ordem exponencial. Então, $L(f'(t)) = s\,F(s) - f(0)$, onde $L(f(t)) = F(s)$.

Teorema (Integração de f(t)) Seja $f(t)$ e $f'(t)$ contínuas para $0 \le t$, e de ordem exponencial . Então, $L(\int f(t)\,dt) = \frac{F(s)}{s}$, onde $L(f(t)) = F(s)$.

Vamos ver alguns exemplos de como usar estes resultados.

Carregue os procedimentos para transformada de Laplace. Faça isto com.

```
> with(inttrans):
```

Exemplo 1 Determine $L(\cos(t)^2)$.

```
> f:='f': F:='F': s:='s': t:='t': T:='T':
f := t -> cos(t)^2:
f1 := t -> subs(T=t,diff(f(T),T)):
```

```
`f(t) ` = f(t);
`f(0) ` = f(0);
`f '(t) ` = f1(t);
```

$$f(t) = \cos(t)^2$$

$$f(0) = 1$$

$$f'(t) = -2\cos(t)\sin(t)$$

Note que $\sin(2t) = 2\sin(t)\cos(t)$ e assim use que $L(\sin(2t)) = \frac{2}{s^2+4}$.

```
> LDf := - 2/(s^2 + 4):
LDf = `L(f '(t)) `;
eqn := LDf = s*F(s) - f(0): eqn;
sol := solve(eqn, F(s)):
`Resolva para F(s).`;
`F(s) ` = sol;
sol := simplify(sol):
`F(s) ` = sol;
```

$$-2\frac{1}{s^2+4} = L(f'(t))$$

$$-2\frac{1}{s^2+4} = s\,F(s) - 1$$

Resolva para F(s).

$$F(s) = \frac{2+s^2}{s(s^2+4)}$$

$$F(s) = \frac{2+s^2}{s(s^2+4)}$$

Verifique isto usando as rotinas para Laplace

```
> `f(t) ` = cos(t)^2;
`F(s) ` = laplace(cos(t)^2, t, s);
```

$$f(t) = \cos(t)^2$$

$$F(s) = 2\,\frac{1+\frac{1}{2}s^2}{s\,(s^2+4)}$$

Surpresa, o Maple NÃO pode calcular!

Exemplo 2 Use o teorema acima apra determinar $\mathrm{L}(\mathrm{f}(t))$,onde

(a) $\mathrm{L}(t^2)$.

Como $f\,'(t) = 2\,t$ e $\mathrm{L}(2\,t) = \frac{2}{s^2}$.

```
> f:='f': F:='F': s:='s': t:='t':
f := t -> t^2:
f1 := t -> subs(T=t,diff(f(T),T)):
`f(t) ` = f(t);
`f '(t) ` = f1(t);
LDf := laplace(f1(t), t, s):
`L(f '(t)) ` = LDf;
Lf := LDf/s:
`F(s) = L(f '(t))/s ` = Lf;
```

$$f(t) = t^2$$

$$f\,'(t) = 2\,t$$

$$L(f\,'(t)) = 2\,\frac{1}{s^2}$$

$$F(s) = L(f\,'(t))/s = 2\,\frac{1}{s^3}$$

Verifique isto com as rotinas do Maple para Laplace.

```
> `f(t) ` = t^2;
`F(s) ` = laplace(t^2, t, s);
```

$$f(t) = t^2$$

$$F(s) = 2\frac{1}{s^3}$$

(b) $\mathrm{L}(t^3)$.

Como $f\,'(t) = 3\,t^2$ e $\mathrm{L}(3\,t^2) = \dfrac{6}{s^3}$.

```
> f:='f': F:='F': s:='s': t:='t':
f := t -> t^3:
f1 := t -> subs(T=t,diff(f(T),T)):
`f(t) ` = f(t);
`f '(t) ` = f1(t);
LDf := laplace(f1(t), t, s):
`L(f '(t)) ` = LDf;
Lf := LDf/s:
`F(s) = L(f '(t))/s ` = Lf;
```

$$f(t) = t^3$$

$$f\,'(t) = 3\,t^2$$

$$L(f\,'(t)) = 6\frac{1}{s^3}$$

$$F(s) = L(f\,'(t))/s = 6\frac{1}{s^4}$$

Verifique isto com as rotinas do Maple.

```
> `f(t) ` = t^3;
`F(s) ` = laplace(t^3, t, s);
```

$$f(t) = t^3$$

$$F(s) = 6\frac{1}{s^4}$$

Exemplo 3 Resolver o seguinte problema de valor inicial

$y''(t) + y(t) = 0$ com $y(0) = 2$ e $y'(0) = 3$.

```
> s:='s': t:='t': Y:='Y': Ys:='Ys':
y0 := 2:
y1 := 3:
F := 0:
`y''(t) + y(t) = 0`;
`y(0) ` = y0,` y'(0) ` = y1;
eqn := s^2*Y(s) - s*y0 - y1 + Y(s) = F: eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
```

$$y''(t) + y(t) = 0$$

$$y(0) = 2, \quad y'(0) = 3$$

$$s^2\, Y(s) - 2\, s - 3 + Y(s) = 0$$

$$Y(s) = \frac{2\, s + 3}{s^2 + 1}$$

Usando que $L(\cos(t)) = \frac{s}{s^2+1}$ e $L(\sin(t)) = \frac{1}{s^2+1}$
vamos determinar a solução que é uma combinação linear de cos(t) e sin(t) .

```
> F1 := s/(s^2+1):
F2 := 1/(s^2+1):
`Y(s) ` = 2*F1 + 3*F2;
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`f(t) ` = 2*f1 + 3*f2;
```

$$Y(s) = 2\,\frac{s}{s^2+1} + 3\,\frac{1}{s^2+1}$$

$$f(t) = 2\cos(t) + 3\sin(t)$$

Podemos usar o Maple para determinar diretamente a inversa.

```
> `Y(s) ` = Y(s);
`f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = \frac{2\,s+3}{s^2+1}$$

$$f(t) = 2\cos(t) + 3\sin(t)$$

Podemos usar os procedimentos do Maple para EDO para obter a solução diretamente.

```
> y:='y': t:='t':
ODE := diff(y(t),t$2)+y(t) = 0:
ICs := {y(0)=2, D(y)(0)=3}:
`D. E. ` = ODE;
`I. C.'s ` = ICs;
dsolve(ODE, y(t));
dsolve({ODE} union ICs, y(t));
```

$$D.\ E. = \left(\left(\frac{\partial^2}{\partial t^2}\mathrm{y}(t)\right) + \mathrm{y}(t) = 0\right)$$

$$I.\ C.'s = \{\mathrm{y}(0) = 2, \mathrm{D}(y)(0) = 3\}$$

$$\mathrm{y}(t) = \_C1\,\sin(t) + \_C2\,\cos(t)$$

$$\mathrm{y}(t) = 2\cos(t) + 3\sin(t)$$

Exemplo 4 Resolva o problema de valor inicial

$y''(t) + y'(t) - 2y(t) = 0$ com $\mathrm{y}(0) = 1$ e $y'(0) = 4$.

```
> s:='s': S:='S': t:='t': Y:='Y':
y0 := 1:
y1 := 4:
F := 0:
`y''(t) + y'(t) - 2y(t) = 0`;
`y(0) ` = y0, ` y'(0) ` = y1;
eqn := s^2*Y(s) - s*y0 - y1 + s*Y(s) - y0 - 2*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
```

```
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y''(t) + y'(t) - 2y(t) = 0$$

$$y(0) = 1, \quad y'(0) = 4$$

$$s^2\,Y(s) - s - 5 + s\,Y(s) - 2\,Y(s) = 0$$

$$Y(s) = \frac{s+5}{s^2+s-2}$$

$$Y(s) = -\frac{1}{s+2} + 2\,\frac{1}{s-1}$$

Usando que $L(e^{(-2\,t)}) = \frac{1}{s+2}$ e $L(e^{t}) = \frac{1}{s-1}$
a solução é combinação linear.

```
> F1 := 1/(s+2):
F2 := 1/(s-1):
`Y(s) ` = - F1 + 2*F2;
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`f(t) ` = - f1 + 2*f2;
```

$$Y(s) = -\frac{1}{s+2} + 2\,\frac{1}{s-1}$$

$$f(t) = -e^{(-2\,t)} + 2\,e^{t}$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
`f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = \frac{s+5}{s^2+s-2}$$

$$f(t) = -e^{(-2\,t)} + 2\,e^{t}$$

Podemos verificar isto com o Maple.

```
> y:='y': t:='t':
ODE := diff(y(t),t$2)+diff(y(t),t)-2*y(t) = 0:
ICs := {y(0)=1, D(y)(0)=4}:
`D. E. ` = ODE;
`I. C.'s ` = ICs;
dsolve(ODE, y(t));
dsolve({ODE} union ICs, y(t));
```

$$D.\ E. \quad = \left(\left(\left(\frac{\partial^2}{\partial t^2} y(t)\right) + \left(\frac{\partial}{\partial t} y(t)\right) - 2\,y(t) = 0\right)\right)$$

$$I.\ C.'s \quad = \{y(0) = 1, D(y)(0) = 4\}$$

$$y(t) = \_C1\ e^{t} + \_C2\ e^{(-2\,t)}$$

$$y(t) = -e^{(-2\,t)} + 2\,e^{t}$$

Exemplo 5 Resolva o problema de valor inicial

$y''(t) - 3\,y'(t) + 2y(t) = 0$ com $y(0) = -1$ e $y'(0) = 4$.

```
> s:='s': S:='S': t:='t': Y:='Y':
y0 := -1:
y1 := 4:
F := 0:
`y''(t) -3y'(t) + 2y(t) = 0`;
`y(0) ` = y0, ` y'(0) ` = y1;
eqn := s^2*Y(s) - s*y0 - y1 -3*(s*Y(s) - y0) + 2*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y''(t)\ -3y'(t)\ +\ 2y(t) =\ 0$$

$$y(0)\ = -1,\quad y'(0)\ = 4$$

$$s^2\ Y(s) + s - 7 - 3\,s\ Y(s) + 2\ Y(s) = 0$$

$$Y(s) = -\frac{s-7}{s^2 - 3s + 2}$$

$$Y(s) = -6\frac{1}{s-1} + 5\frac{1}{s-2}$$

Usando que $L(e^t) = \frac{1}{s-1}$ e $L(e^{(2t)}) = \frac{1}{s-2}$
a solução é combinação linear.

```
> F1 := 1/(s-2):
F2 := 1/(s-1):
`Y(s) ` = 5*F1 - 6*F2;
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`f(t) ` = 5* f1 -6*f2;
```

$$Y(s) = -6\frac{1}{s-1} + 5\frac{1}{s-2}$$

$$f(t) = 5\,e^{(2t)} - 6\,e^t$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
`f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = -\frac{s-7}{s^2 - 3s + 2}$$

$$f(t) = 5\,e^{(2t)} - 6\,e^t$$

Podemos verificar isto com o Maple.

```
> y:='y': t:='t':
ODE := diff(y(t),t$2)-3*diff(y(t),t)+2*y(t) = 0:
ICs := {y(0)=-1, D(y)(0)=4}:
`D. E. ` = ODE;
`I. C.'s ` = ICs;
dsolve(ODE, y(t));
dsolve({ODE} union ICs, y(t));
```

$$D.E. := \left(\left(\frac{\partial^2}{\partial t^2}y(t)\right) - 3\left(\frac{\partial}{\partial t}y(t)\right) + 2\,y(t) = 0\right)$$

$$I.C.'s := \{D(y)(0) = 4, y(0) = -1\}$$

$$y(t) = \_C1\,\mathbf{e}^t + \_C2\,\mathbf{e}^{(2\,t)}$$

$$y(t) = 5\,\mathbf{e}^{(2\,t)} - 6\,\mathbf{e}^t$$

Teoremas de Deslocamento

Teorema (deslocamento na variável s) Se $F(s)$ é a transformada de Laplace de $f(t)$, então

$$L(\mathbf{e}^{(a\,t)}\,f(t)) = F(s-a)$$

.

Teorema (deslocamento na variável t) Se $F(s)$ é a transformada de Laplace de $f(t)$ e $0 \le a$, então $L(U_a(t)\,f(t-a)) = \mathbf{e}^{(-a\,s)}\,F(s)$.

Carregue o pacote de transformações para trabalhar com o Maple.

```
> with(inttrans):
```

$$L(t^n e^{(a t)})$$

Exemplo 1 Calcule

Usando que $L(t^n) = \frac{n!}{s^{(n+1)}}$. e fazemos o deslocamento.

```
> a:='a': f:='f': F:='F': n:='n': s:='s': t:='t':
f := t -> t^n:
F := s -> n!/s^(n+1):
`formulas dadas:`;
`f(t) ` = f(t);
`F(s) ` = F(s);
`deslocamento na variavel s para obter:`;
`f(t) ` = f(t)*exp(a*t);
`F(s) ` = F(s-a);
```

$$\mathit{formulas\ dadas:}$$

$$f(t) = t^n$$

$$F(s) = \frac{n!}{s^{(n+1)}}$$

$$\mathit{deslocamento\ na\ variavel\ s\ para\ obter:}$$

$$f(t) = t^n e^{(a t)}$$

$$F(s) = \frac{n!}{(-a+s)^{(n+1)}}$$

Podemos verificar isto com o Maple.

```
> assume(A,positive);
assume(N,positive);
`Por exemplo, comece com:`;
`f(t) ` = t^n;
L := laplace(t^N, t, s):
`F(s) ` = subs(N='n',L);
`Deslocamento na variável s para obter:`;
`f(t) ` = t^n*exp(a*t);
LS := laplace(t^N*exp(A*t), t, s):
`F(s) ` = subs({A='a',N='n'},LS);
```

Por exemplo, comece com:

$$f(t) = t^n$$

$$F(s) = s^{(-n-1)}\,\Gamma(n+1)$$

Deslocamento na variável s para obter:

$$f(t) = t^n\,\mathbf{e}^{(a\,t)}$$

$$F(s) = \Gamma(n+1)\,(s-a)^{(-n-1)}$$

Exemplo 2 Resolva o PVI

$y''(t) + y(t) = U_\pi(t)$ com $y(0) = 0$ e $y'(0) = 0$

```
> s:='s': S:='S': t:='t': Y:='Y': Y:='Y':
y0 := 0:
y1 := 0:
F := laplace(Heaviside(t-Pi), t, s):
`y''(t) + y(t) = UPi(t)`;
`y(0) ` = y0,` y'(0) ` = y1;
eqn := s^2*Y(s) - s*y0 - y1 + Y(s) = F: eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
1/(s*(s^2+1)) = convert(1/(s*(s^2+1)), parfrac, s);
`Y(s) ` = exp(-Pi*s)/s - exp(-Pi*s)*s/(s^2+1);
```

$$y''(t) \;+\; y(t) \;=\; UPi(t)$$

$$y(0) = 0, \quad y'(0) = 0$$

$$s^2\,\mathrm{Y}(s) + \mathrm{Y}(s) = \frac{\mathbf{e}^{(-\pi\,s)}}{s}$$

$$Y(s) = \frac{\mathbf{e}^{(-\pi\,s)}}{s\,(s^2+1)}$$

$$\frac{1}{s\,(s^2+1)} = \frac{1}{s} - \frac{s}{s^2+1}$$

$$Y(s) = \frac{e^{(-\pi s)}}{s} - \frac{e^{(-\pi s)}\,s}{s^2+1}$$

Use que $\mathrm{L}(U_\pi(t)) = \dfrac{e^{(-\pi s)}}{s}$ e $\mathrm{L}(U_\pi(t)\cos(t-\pi)) = \dfrac{e^{(-\pi s)}\,s}{s^2+1}$.

```
> F1 := exp(-Pi*s)/s:
F2 := exp(-Pi*s)*s/(s^2+1):
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`Y(s) ` = F1 - F2;
`f(t) ` = f1 - f2;
```

$$Y(s) = \frac{e^{(-\pi s)}}{s} - \frac{e^{(-\pi s)}\,s}{s^2+1}$$

$$f(t) = \mathrm{Heaviside}(t-\pi) + \mathrm{Heaviside}(t-\pi)\cos(t)$$

Podemos verificar isto usando o Maple.

```
> ODE := diff(y(t),t$2)+y(t) = Heaviside(t-Pi):
ICs := {y(0)=0, D(y)(0)=0}:
`D. E. ` = ODE;
`I. C.'s ` = ICs;
dsolve(ODE, y(t), method=laplace);
dsolve({ODE} union ICs, y(t), method=laplace);
```

$$D.\ E. = \left(\left(\frac{\partial^2}{\partial t^2}\mathrm{y}(t)\right) + \mathrm{y}(t) = \mathrm{Heaviside}(t-\pi)\right)$$

$$I.\ C.'s = \{\mathrm{y}(0)=0, \mathrm{D}(y)(0)=0\}$$

$$\mathrm{y}(t) = \mathrm{y}(0)\cos(t) + \mathrm{D}(y)(0)\sin(t) + 2\,\mathrm{Heaviside}(t-\pi)\cos\left(\frac{1}{2}t\right)^2$$

$$y(t) = 2\,\mathrm{Heaviside}(t-\pi)\cos\left(\frac{1}{2}t\right)^2$$

Invertendo a Transformada de Laplace

```
> with(inttrans):
laplace(t,t,s):
```

Exemplo 1 Determine a transrformada inversa de $Y(s) = \dfrac{s^3 - 4s + 1}{s(s-1)^3}$.

```
> s:='s': Y:='Y':
Y := s ->(s^3 - 4*s + 1)/(s*(s-1)^3):
`Y(s) ` = Y(s);
```

$$Y(s) = \frac{s^3 - 4s + 1}{s(s-1)^3}$$

Vamos usar frações parciais para decompor a expressão de Y(s) em fatores mais simples.

```
> s:='s': Y:='Y':
Y := s ->(s^3 - 4*s + 1)/(s*(s-1)^3):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s),parfrac,s);
```

$$Y(s) = \frac{s^3 - 4s + 1}{s(s-1)^3}$$

$$Y(s) = -\frac{1}{s} - 2\frac{1}{(s-1)^3} + \frac{1}{(s-1)^2} + 2\frac{1}{s-1}$$

Usamos a tabela de transformada de Laplace para procurar a inversa, obtemos a solução

$$f(t) = -1 - t^2\,e^t + t\,e^t + 2\,e^t$$

.

Podemos usar o Maple para conferir.

```
> `F(s) ` = (s^3 - 4*s + 1)/(s*(s-1)^3);
`f(t) ` = invlaplace((s^3 - 4*s + 1)/(s*(s-1)^3), s, t);
```

$$F(s) = \frac{s^3 - 4s + 1}{s(s-1)^3}$$

$$f(t) = -1 - t^2 e^t + t e^t + 2 e^t$$

Exemplo 2 Determine a transformada inversa de $\mathrm{F}(s) = \dfrac{5s}{(s^2+4)(s^2+9)}$.

```
> s:='s': Y:='Y':
Y := s -> 5*s/((s^2+4)*(s^2+9)):
`Y(s) ` = Y(s);
```

$$Y(s) = 5\frac{s}{(s^2+4)(s^2+9)}$$

Podemos usar o Maple para converter em frações parciais.

```
> s:='s': Y:='Y':
Y := s -> 5*s/((s^2+4)*(s^2+9)):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s),parfrac,s);
```

$$Y(s) = 5\frac{s}{(s^2+4)(s^2+9)}$$

$$Y(s) = \frac{s}{s^2+4} - \frac{s}{s^2+9}$$

Use a tabela de transformadad e Laplace para determinar a inversa:

$\mathrm{y}(t) = \cos(2t) - \cos(3t)$

.

Exemplo 3 Determine a transformada inversa de $\mathrm{Y}(s) = \frac{s^3 + 3s^2 - s + 1}{s(s+1)^2(s^2+1)}$.
Em frações parciais, temos

```
> s:='s': Y:='Y':
Y := s ->(s^3+3*s^2-s+1)/(s*(s+1)^2*(s^2+1)):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s),parfrac,s);
```

$$Y(s) = \frac{s^3 + 3s^2 - s + 1}{s(s+1)^2(s^2+1)}$$

$$Y(s) = \frac{1}{s} - 2\frac{1}{(s+1)^2} - 2\frac{1}{s+1} + \frac{s+1}{s^2+1}$$

Usando uma tabela de transformada de Laplace temos que a resposta é:

$$\mathrm{y}(t) = 1 - 2t\,\mathbf{e}^{(-t)} - 2\,\mathbf{e}^{(-t)} + \cos(t) + \sin(t)$$

.

Exemplos de PVI

```
> with(inttrans):
```

Exemplo 1 Resolva o problema de valor inicial

$y''(t) - 3y'(t) + 2y(t) = 0$ com $\mathrm{y}(0) = 3$ e $y'(0) = 4$.

```
> s:='s': S:='S': t:='t': Y:='Y':
y0 := 3:
y1 := 4:
F := 0:
`y''(t) -3y'(t) + 2y(t) = 0`;
`y(0) ` = y0, ` y'(0) ` = y1;
eqn := s^2*Y(s) - s*y0 - y1 -3*(s*Y(s) - y0) + 2*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y''(t)\ -3y'(t)\ +\ 2y(t) =\ 0$$

$$y(0)\ = 3,\quad y'(0)\ = 4$$

$$s^2\,\mathrm{Y}(s) - 3\,s + 5 - 3\,s\,\mathrm{Y}(s) + 2\,\mathrm{Y}(s) = 0$$

$$Y(s)\ = \frac{3\,s-5}{s^2-3\,s+2}$$

$$Y(s)\ = 2\,\frac{1}{s-1}+\frac{1}{s-2}$$

Usando que $\mathrm{L}(\mathbf{e}^t) = \dfrac{1}{s-1}$ e $\mathrm{L}(\mathbf{e}^{(2\,t)}) = \dfrac{1}{s-2}$

a solução é combinação linear.

```
> F1 := 1/(s-2):
F2 := 1/(s-1):
`Y(s) ` = F1 + 2*F2;
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`f(t) ` = f1 + 2*f2;
```

$$Y(s)\ = 2\,\frac{1}{s-1}+\frac{1}{s-2}$$

$$f(t)\ = \mathbf{e}^{(2\,t)} + 2\,\mathbf{e}^{t}$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
`f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s)\ = \frac{3\,s-5}{s^2-3\,s+2}$$

$$f(t)\ = \mathbf{e}^{(2\,t)} + 2\,\mathbf{e}^{t}$$

Podemos verificar isto com o Maple.

```
> y:='y': t:='t':
ODE := diff(y(t),t$2)-3*diff(y(t),t)+2*y(t) = 0:
```

```
ICs := {y(0)=3, D(y)(0)=4}:
`D. E. ` = ODE;
`I. C.'s ` = ICs;
dsolve(ODE, y(t));
dsolve({ODE} union ICs, y(t));
```

$$D.\ E. = \left(\left(\frac{\partial^2}{\partial t^2} y(t) \right) - 3 \left(\frac{\partial}{\partial t} y(t) \right) + 2\, y(t) = 0 \right)$$

$$I.\ C.'s = \{ y(0) = 3, D(y)(0) = 4 \}$$

$$y(t) = \_C1\, e^{t} + \_C2\, e^{(2\,t)}$$

$$y(t) = e^{(2\,t)} + 2\, e^{t}$$

Exemplo 2 Resolva o problema de valor inicial

$y''(t) - 3\, y'(t) + 2y(t) = 0$ com $y(0) = -1$ e $y'(0) = 4$.

```
> s:='s': S:='S': t:='t': Y:='Y':
y0 := -1:
y1 := 4:
F := 0:
`y''(t) -3y'(t) + 2y(t) = 0`;
`y(0) ` = y0, ` y'(0) ` = y1;
eqn := s^2*Y(s) - s*y0 - y1 -3*(s*Y(s) - y0) + 2*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y''(t)\ -3y'(t)\ +\ 2y(t) =\ 0$$

$$y(0)\ = -1,\quad y'(0)\ = 4$$

$$s^2\, Y(s) + s - 7 - 3\, s\, Y(s) + 2\, Y(s) = 0$$

$$Y(s)\ = -\frac{s-7}{s^2 - 3\, s + 2}$$

$$Y(s) = -6\frac{1}{s-1} + 5\frac{1}{s-2}$$

Usando que $L(e^t) = \frac{1}{s-1}$ e $L(e^{(2t)}) = \frac{1}{s-2}$
a solução é combinação linear .

```
> F1 := 1/(s-2):
F2 := 1/(s-1):
`Y(s) ` = -6* F1 + 5*F2;
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`f(t) ` = - 6*f1 + 5*f2;
```

$$Y(s) = -6\frac{1}{s-2} + 5\frac{1}{s-1}$$

$$f(t) = -6\,e^{(2t)} + 5\,e^t$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
`f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = -\frac{s-7}{s^2-3s+2}$$

$$f(t) = 5\,e^{(2t)} - 6\,e^t$$

Podemos verificar isto com o Maple.

```
> y:='y': t:='t':
ODE := diff(y(t),t$2)-3*diff(y(t),t)+2*y(t) = 0:
ICs := {y(0)=-1, D(y)(0)=4}:
`D. E. ` = ODE;
`I. C.'s ` = ICs;
dsolve(ODE, y(t));
dsolve({ODE} union ICs, y(t));
```

$$D.\ E. = \left(\left(\frac{\partial^2}{\partial t^2}y(t)\right) - 3\left(\frac{\partial}{\partial t}y(t)\right) + 2\,y(t) = 0\right)$$

$$I.C.'s \ = \{y(0) = -1, D(y)(0) = 4\}$$

$$y(t) = \_C1\, e^{t} + \_C2\, e^{(2\,t)}$$

$$y(t) = 5\, e^{(2\,t)} - 6\, e^{t}$$

```
> restart:
```

```
> with(inttrans):
```

Exemplo 3 Resolva o problema de valor inicial

$y''(t) + y(t) = t$ com $y(0) = -1$ e $y'(0) = 3$.

```
> s:='s': S:='S': t:='t': Y:='Y':
y0 := -1:
y1 := 3:
F := laplace(t, t , s):
`y''(t) + y(t) = t`;
`y(0) ` = y0, ` y'(0) ` = y1;
eqn := s^2*Y(s) - s*y0 - y1 +1*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y''(t) + y(t) = t$$

$$y(0) = -1, \quad y'(0) = 3$$

$$s^2\, Y(s) + s - 3 + Y(s) = \frac{1}{s^2}$$

$$Y(s) = -\frac{s^3 - 3\, s^2 - 1}{s^2\,(s^2 + 1)}$$

$$Y(s) = \frac{1}{s^2} - \frac{-2 + s}{s^2 + 1}$$

Usando que $L(t) = \frac{1}{s^2}$ e $L(2\sin(t) - \cos(t)) = \frac{s-2}{s^2+1}$

a solução é combinação linear.

```
> F1 := 1/(s^2):
F2 := (s-2)/(s^2+1):
`Y(s) ` = F1 -F2;
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`f(t) ` = f1 -f2;
```

$$Y(s) = \frac{1}{s^2} - \frac{-2+s}{s^2+1}$$

$$f(t) = t - \cos(t) + 2\sin(t)$$

```
>

> restart:

> with(inttrans):
```

Exemplo 4 Resolva o problema de valor inicial

$y''(t) + y'(t) - y(t) = 4*exp(t)$ com $y(0) = 1$ e $y'(0) = 0$.

```
> s:='s': S:='S': t:='t': Y:='Y':
y0 := 1:
y1 := 0:
F := laplace(4*exp(t), t , s);
`y''(t) + y'(t)-y(t) = 4*exp(t)`;
`y(0) ` = y0, ` y'(0) ` = y1;
eqn := s^2*Y(s) - s*y0 - y1 +(s*Y(s) - y0)-1*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$F := 4\frac{1}{s-1}$$

$$y''(t) + y'(t) - y(t) = 4*exp(t)$$

$$y(0) = 1, \quad y'(0) = 0$$

$$s^2\, Y(s) - s - 1 + s\, Y(s) - Y(s) = 4\, \frac{1}{s-1}$$

$$Y(s) = \frac{s^2+3}{s^3 - 2\,s + 1}$$

$$Y(s) = 4\, \frac{1}{s-1} - \frac{7+3\,s}{s^2+s-1}$$

Algo mais sobre EDO's

Métodos Numéricos para EDO no Maple

```
> restart:
```

Se você está interessado na solução de um problema de valor inicial em um ponto particular ou em um número pequeno de pontos particulares, é mais eficiente resolver numericamente a equação diferencial do que tentar resolvê-la exatamente usando fórmulas. Muitas vezes, não existe uma expressão explícita para a solução, e aí teremos que usar métodos numéricos.

Vamos ver agora como se pode obter a solução numérica usando o Maple.

O comando dsolve, sem informação de opção tentará uma solução exata. Uusando o comando dsolve com a opção "numeric" o Maple retorna um procedimento que você deverá usar para obter a solução no ponto procurado.

Como default o Maple usa o método de Runge-Kutta de ordem 4.

Exemplo 1: Considere a EDO y'= $x^2 - y^2$, com y(0)=1.

```
> {D(y)(x) = x^2 - y^2, y(0)=1};
```

$$\{D(y)(x) = x^2 - y^2, y(0) = 1\}$$

```
> solucao:=dsolve({D(y)(x) = x^2 - y(x)^2, y(0)=1}, y(x), numeric);
```

$$solucao := \mathbf{proc}(rkf45\_x)\ \ldots\ \mathbf{end}$$

O Maple retornou um procedimento que dará a solução no ponto desejado. Para saber a solução no ponto

x=1, execute a linha abaixo.

```
> solucao(1); #dá a solução no ponto x=1 e etc.
```

$$[x = 1, y(x) = .7500156981235434]$$

```
> solucao(2);
```

$$[x = 2, y(x) = 1.679458970842937]$$

```
> solucao(5);
```

$$[x = 5, y(x) = 4.896724060220174]$$

Não temos nenhum controle, deixando o Maple fazer todas as contas e tomar todas as decisões. Para obter mais controle devemos especificar o método a ser usado pelo Maple e também o passo a ser usado. Existem três métodos numéricos que usamos mais comumente: método de Euler, Método de Heum e o Runge-Kutta.

Vamos ver como isto pode ser feito.

Exemplo 2 : Considere a seguinte EDO y' = $y^2 - x$ com y(0)=0.

```
> edo := D(y)(x) = y(x)^2-x;
```

$$edo := D(y)(x) = y(x)^2 - x$$

Estimar y(1) se y(0)=0, usando o método de Euler com passo h = 0.01.

```
> eulersol:=dsolve({edo, y(0)=0}, y(x), numeric,
method=classical[foreuler], stepsize=0.01);
```

$$eulersol := \mathbf{proc}(x\_classical) \;\ldots\; \mathbf{end}$$

```
> eulersol(1);
```

$$[x = 1, y(x) = -.4524652340061601]$$

Estimar y(1) se y(0)=0, usando o método de Euler melhorado com passo h = 0.05.

> **Melhoreulersol:=dsolve({edo, y(0)=0}, y(x), numeric, method=classical[heunform], stepsize=0.05);**

$$Melhoreulersol := \mathbf{proc}(x\_classical) \ \ldots \ \mathbf{end}$$

> **Melhoreulersol(1);**

$$[x = 1, y(x) = -.4555263344850716]$$

Estimar y(1) se y(0)=0, usando o método de Runge-Kutta com passo h = 0.1.

> **rksol:=dsolve({edo, y(0)=0}, y(x), numeric, method=classical[rk4], stepsize=0.1);**

$$rksol := \mathbf{proc}(x\_classical) \ \ldots \ \mathbf{end}$$

> **rksol(1);**

$$[x = 1, y(x) = -.4555438427043770]$$

Exemplo 3

Vamos ver agora como usar os recursos numéricos do Maple para plotar campos de direções de EDO's.

Uma EDO y'(x)= f(x, y) define um campo de direções. O Maple pode plotar este campo. Vejamos como se se faz isto.

Antes precisamos carregar o pacote de ferramentas para equações diferenciais.

> **with(DEtools):**

> **eqc1:= diff(y(x),x) + y(x) = 0;**

$$eqc1 := \left(\frac{\partial}{\partial x} y(x)\right) + y(x) = 0$$

> **ODE:= diff(y(x),x) = sqrt(y(x)^2+1);**

$$ODE := \frac{\partial}{\partial x} y(x) = \sqrt{y(x)^2 + 1}$$

> **DEplot(eqc1,y(x),x=-4..4,y=-4..4,title=`Exemplo 1`, arrows=slim);**



> **DEplot(ODE,y(x),x=-4..4,y=-4..4,title=`Exemplo 2`, arrows=slim);**



> **eqc2 := diff(y(x),x) = (y(x)-x)*(1-y(x)^3);**

$$eqc2 := \frac{\partial}{\partial x}\,y(x) = (y(x) - x)\,(1 - y(x)^3)$$

Gerando um campo de direções

> **DEplot(eqc2,y(x),x=0..2,y=0.5..2,title=`Exemplo 3`,arrows=SLIM);**



Podemos também plotar a solução usando o método de Euler.

```
> DEplot(eqc2,y(x),x=0..2,{[0,2],[0,0.5]},
arrows=NONE,method=classical[foreuler]);
```



Usando método de Euler melhorado.

```
> DEplot(eqc2,y(x),x=0..2,{[0,2],[0,0.5]},
arrows=NONE,method=classical[heunform]);
```



Usando Runge-Kutta4, default do Maple.

> **DEplot(eqc2,y(x),x=0..2,{[0,2],[0,0.5]},arrows=SMALL);**



> **DEplot(eqc2,y(x),x=0..2,{[0,2],[0,0.5]},
arrows=NONE,method=classical[heunform]);**



Alguns Modelos

> **restart:**

> **with(inttrans):**

Modelo 1:Desintegração Radioativa

Se a quantidade do material radiotivo no instante t é Q(t),
então a taxa de variação no instante t é proporcional a
quantidade Q(t), isto é,

$$\frac{d\,Q(t)}{dt} = -k\,Q(t)$$

,
onde k>0 é chamada a constante de desintegração radiotiva do
material.

Se a quantidade inicial de material é Q[0], então podemos usar a transformada
de Laplace para determinarmos a solução do PVI:

$$\frac{d\,Q(t)}{dt} = -k\,Q(t)$$

Q(0)= Q_0

```
> s:='s': S:='S': t:='t': Y:='Y':
y0 := Q[0]:
F := 0:
`y'(t) +ky(t) = 0`;
```

```
`y(0) ` = y0;
eqn := s*Y(s) - y0 +k*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y'(t) \; +ky(t) = 0$$

$$y(0) = Q_0$$

$$s\,Y(s) - Q_0 + k\,Y(s) = 0$$

$$Y(s) = \frac{Q_0}{s+k}$$

$$Y(s) = \frac{Q_0}{s+k}$$

$$L(e^{(-kt)}) = \frac{1}{s+k}$$

Usando que
a solução é combinação linear .

```
> F1 := 1/(s+k):
`Y(s) ` = Q[0]* F1;
f1 := invlaplace(F1, s , t):
`f(t) ` = Q[0]*f1 ;

>
```

$$Y(s) = \frac{Q_0}{s+k}$$

$$f(t) = Q_0\, e^{(-k\,t)}$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
`f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = \frac{Q_0}{s+k}$$

$$f(t) = Q_0 \, e^{(-k\,t)}$$

Modelo 2: Movimento num meio viscoso

Uma partícula de massa m é abandonada num meio viscoso, isto é, num meio que oferece resistência ao seu movimento proporcional a velocidade. O movimento da partícula no meio viscoso depende da constante k de viscosidade, da seguinte forma

F= $m\,a = m\,g - k\,v$,

onde a é a aceleração e v é a velocidade.

Reescrevendo,

$$\frac{dv}{dt} + k\,m\,v = g$$

.

Supondo que $v(0) = v_0$, obtemos a solução via Laplace dada por

```
> restart:

> with(inttrans):

> s:='s': S:='S': t:='t': Y:='Y':
y0 := v[0]:
F := g/s:
`y'(t) +kmy(t) = g`;
`y(0) ` = y0;
eqn := s*Y(s) - y0 +k*m*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y'(t) \;\; +kmy(t) = \; g$$

$$y(0) \; = v_0$$

$$s\,\mathrm{Y}(s) - v_0 + k\,m\,\mathrm{Y}(s) = \frac{g}{s}$$

$$Y(s) = \frac{v_0\,s + g}{s\,(s + k\,m)}$$

$$Y(s) = \frac{g}{k\,m\,s} + \frac{k\,m\,v_0 - g}{k\,m\,(s + k\,m)}$$

Usando que $\mathrm{L}(\mathbf{e}^{(-kmt)}) = \dfrac{1}{s + km}$ e $\mathrm{L}(1) = \dfrac{1}{s}$
a solução é combinação linear.

```
> F1 := 1/(s+k*m):
F2 := 1/(s):
`Y(s) ` = (k*m*v[0]-g)/(k*m)* F1+ (g/k*m)*F2;
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`f(t) ` = (k*m*v[0]-g)/(k*m)*f1 + (g/k*m)*f2 ;
```

$$Y(s) = \frac{k\,m\,v_0 - g}{k\,m\,(s + k\,m)} + \frac{g\,m}{k\,s}$$

$$f(t) = \frac{(k\,m\,v_0 - g)\,\mathbf{e}^{(-k\,m\,t)}}{k\,m} + \frac{g\,m}{k}$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
`f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = \frac{v_0\,s + g}{s\,(s + k\,m)}$$

$$f(t) = \frac{g}{k\,m} + \mathbf{e}^{(-k\,m\,t)}\,v_0 - \frac{\mathbf{e}^{(-k\,m\,t)}\,g}{k\,m}$$

Modelo 3: Lei de resfriamento de Newton

A lei de resfriamento de Newton diz que a taxa de variação da

temperatura $\frac{dT}{dt}$ de um corpo em relação ao
tempo é proporcional a diferença da sua temperatura T e da
temperatura ambiente T_0 , isto é,

$$\frac{dT}{dt} = -k\,(T - T_0)$$

,

onde k>0 é uma constante que depende do material.

.

Supondo que $T(0) = C$, obtemos a solução via Laplace dada por

```
> restart:

> with(inttrans):

> s:='s': S:='S': t:='t': Y:='Y':
y0 := C:
F := k*T[0]/s:
`y'(t) +ky(t) = k*T[0]`;
`y(0) ` = y0;
eqn := s*Y(s) - y0 +k*Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y'(t)\ \ +ky(t) = k*T[0]$$

$$y(0)\ = C$$

$$s\,Y(s) - C + k\,Y(s) = \frac{k\,T_0}{s}$$

$$Y(s)\ = \frac{C\,s + k\,T_0}{s\,(s + k)}$$

$$Y(s)\ = \frac{T_0}{s} + \frac{-T_0 + C}{s + k}$$

Usando que $L(e^{(-k\,t)}) = \dfrac{1}{s+k}$ e $L(1) = \dfrac{1}{s}$
a solução é

```
> F1 := 1/(s):
F2 := 1/(s+k):`Y(s) ` = T[0]*F1+(-T[0]+C)*F2;
f1 := invlaplace(F1, s , t):
f2 := invlaplace(F2, s , t):
`f(t) ` = T[0]*f1+(-T[0]+C)*f2 ;
```

$$Y(s) = \frac{T_0}{s} + \frac{-T_0 + C}{s+k}$$

$$f(t) = T_0 + (-T_0 + C)\, e^{(-k\,t)}$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
`f(t) ` = invlaplace(Y(s), s , t);
```

$$Y(s) = \frac{C\,s + k\,T_0}{s\,(s+k)}$$

$$f(t) = T_0 - e^{(-k\,t)}\, T_0 + e^{(-k\,t)}\, C$$

Modelo 4 : Pêndulo

Diagrama do Pendulo



O diagrama mostra um pêndulo consistindo de uma pequena massa esférica suspensa por um fio. O pêndulo está imerso em um líquido que resiste ao seu movimento. A flecha verde mostra a velocidade v da esfera, a flecha vermelha mostra a força agindo sobre ela.

Existe uma força para baixo g devido à gravidade e também r uma força resistiva devido ao fluido, agindo na direção oposta a velocidade do pêndulo.

A equação que descreve o movimento do pêndulo em termos do ângulo θ entre a esfera do pêndulo e a vertical é dada por:

$$\theta_{tt} = -k\,\theta_t - \theta(t)$$

.

A constante k é a medida que mostra quanta resistência o fluido exerce sobre a esfera. Se k = 0 então o fluido não exerce resistência.

Tomando as condições iniciais dadas por

$\theta(0) = \phi$ e $\theta_t(0) = 0$

obtemos via transformada de Laplace aa seguinte solução.

```
> restart:
```

```
> with(inttrans):

> s:='s': S:='S': t:='t': Y:='Y':
y0 := phi:
y1 := 0:F := 0:
`y''(t) +ky'(t)+y(t) = 0`;
`y(0) ` = y0, ` y'(0) ` = y1;
eqn := s^2*Y(s) -s*y0-y1 +k*(s*Y(s)-y0)+Y(s) = F:
eqn;
sol := simplify(solve(subs(s=S,eqn),Y(S))):
Y := s -> subs(S=s,sol):
`Y(s) ` = Y(s);
`Y(s) ` = convert(Y(s), parfrac, s);
```

$$y''(t)\ \ +ky'(t)+y(t) = 0$$

$$y(0)\ = \phi,\quad y'(0)\ = 0$$

$$s^2\,Y(s) - s\,\phi + k\,(s\,Y(s) - \phi) + Y(s) = 0$$

$$Y(s)\ = \frac{\phi\,(s+k)}{s^2 + k\,s + 1}$$

$$Y(s)\ = \frac{\phi\,(s+k)}{s^2 + k\,s + 1}$$

Podemos usar a inversa de Laplace diretamente.

```
> `Y(s) ` = Y(s);
`f(t) ` = simplify(invlaplace(Y(s), s , t));
```

$$Y(s)\ = \frac{\phi\,(s+k)}{s^2 + k\,s + 1}$$

$$f(t)\ = -\frac{\phi\,\mathbf{e}^{\left(-\frac{1}{2}k\,t\right)}\left(4\cos\left(\frac{1}{2}\sqrt{4-k^2}\,t\right) - k^2\cos\left(\frac{1}{2}\sqrt{4-k^2}\,t\right) + \sqrt{4-k^2}\,k\sin\left(\frac{1}{2}\sqrt{4-k^2}\,t\right)\right)}{-4+k^2}$$

```
>
```